
Automatically Converting Character Variables 
That Store Numbers to Numeric Variables 

ABSTRACT 
A character variable can store numbers, but many analyses require numeric variables. This paper shows how to 
convert character variables that store numbers to numeric variables. The DATA step, PROC SQL, and the SAS® 
macro language are used to automate the process. The result is a new data set that stores all variables with the 
same names, labels, and order, but with character variables that stored numbers converted to numeric variables. 

INTRODUCTION 
Numbers can be stored in numeric variables or character variables. For example, a dummy variable that takes 
the values 1 and 0 could be numeric or character. SAS® will even perform an automatic character-to-numeric 
conversion as needed in specific situations (e.g., when a character variable is compared to a numeric variable). 

System resources, however, are required to convert variables. It is not efficient to have SAS perform automatic 
conversions. Moreover, if SAS tries to convert a character value that contains a non-numeric value to a number 
it produces an error. In addition, the majority of statistical procedures require analysis variables to be numeric. 

Character variables that store numbers and that are treated like numeric variables can be explicitly converted 
to numeric variables. 

SAMPLE DATA 
SAS data set EARNINGS is used in this paper. It contains six variables: ID, EARNQ1 through EARNQ4 (four 
quarterly earnings variables, in dollars), and EMPLOYED (a dummy variable for any earnings in quarters 1-4): 
 

CONTENTS of data set EARNINGS 
 
#    Variable    Type    Len    Label 
 
1    id          Char      2    Unique identifier     
2    earnq1      Num       8    Earnings in Quarter 1 
3    earnq2      Num       8    Earnings in Quarter 2 
4    earnq3      Char      4    Earnings in Quarter 3 
5    earnq4      Char      5    Earnings in Quarter 4 
6    employed    Char      3    Ever employed 

 

Note that ID, EARNQ3, EARNQ4, and EMPLOYED are character variables. 
 

PRINT of data set EARNINGS 
 
Obs    id    earnq1    earnq2    earnq3    earnq4   employed 
 
 1     01     1000      1300      1400     1250       1     
 2     02        0      2000               2300       1     
 3     03        0         0      0        0          UNK     
 4     04     1400      1400      1500     900.0      1     
 5     05        0         .      0        0          UNK 

 

ID values contain leading zeroes that should be preserved. EMPLOYED contains the character value ‘UNK’ (for 
‘Unknown’). EARNQ3 and EARNQ4, however, store only numbers and could be converted to numeric variables. 

CONVERTING VARIABLES 

The following steps are required to convert character variables that store numbers to numeric variables: 

1. Identifying character variables to convert 
2. Creating equivalent numeric variables 
3. Dropping the original character variables 
4. Renaming the equivalent numeric variables 
5. Reordering variables to the original sequence 
6. Labeling the equivalent numeric variables 
 
The required statements are hard-coded in the next section to illustrate this process. The required statements 
are then generated by SAS in the subsequent section to automate this process. 

CONVERTING VARIABLES MANUALLY 
 

STEP 1: IDENTIFYING CHARACTER VARIABLES TO CONVERT 

Use PROC FREQ to produce frequency tables for all character variables: 

proc freq data=earnings; 
tables _character_; 
run; 

The TABLES statement uses the special SAS name list _CHARACTER_ to reference all character variables. 

Inspect the results for non-numeric values. If a variable contains only numeric values, it could be converted 
to a numeric variable: 
 

The FREQ Procedure 
 
                   Unique identifier 
  
                               Cumulative    Cumulative 
id    Frequency     Percent     Frequency      Percent 
------------------------------------------------------- 
01           1       20.00             1        20.00   
02           1       20.00             2        40.00   
03           1       20.00             3        60.00   
04           1       20.00             4        80.00   
05           1       20.00             5       100.00   
 
 
                   Earnings in Quarter 3 
  
                                   Cumulative    Cumulative 
earnq3    Frequency     Percent     Frequency      Percent 
----------------------------------------------------------- 
0                2       50.00             2        50.00   
1400             1       25.00             3        75.00   
1500             1       25.00             4       100.00   
 
Frequency Missing = 1 
 

(continued) 

 

ID contains no non-numeric values. It is a unique identifier, however, and its leading zeroes should be preserved. 
EARNQ3 contains only numeric values (plus a missing value). It could be converted to a numeric variable.  

(continued) 
 
                   Earnings in Quarter 4 
  
                                   Cumulative    Cumulative 
earnq4    Frequency     Percent     Frequency      Percent 
----------------------------------------------------------- 
0                2       40.00             2        40.00   
1250             1       20.00             3        60.00   
2300             1       20.00             4        80.00   
900.0            1       20.00             5       100.00   
 
 
                        Ever employed 
  
                                     Cumulative    Cumulative 
employed    Frequency     Percent     Frequency      Percent 
------------------------------------------------------------- 
1                  3       60.00             3        60.00   
UNK                2       40.00             5       100.00   

 

EARNQ4 also contains only numeric values (including a value with a decimal point). It could be converted to a 
numeric variable. EMPLOYED contains the value ‘UNK’. It could not be converted to a numeric variable. 
 

STEP 2: CREATING EQUIVALENT NUMERIC VARIABLES 

Use the INPUT function in an assignment statement to convert a character variable to a numeric variable: 

x1=input(earnq3,best12.); 
x2=input(earnq4,best12.); 

The INPUT function reads each value of character variable EARNQ3 with the numeric informat BEST12., 
converts it to a numeric value, and stores it in numeric variable X1. The same applies to EARNQ4 and X2. 
 
STEP 3: DROPPING THE ORIGINAL CHARACTER VARIABLES 

Use the DROP statement to drop the original character variables that stored numbers: 

drop earnq3 earnq4; 

 

STEP 4: RENAMING THE EQUIVALENT NUMERIC VARIABLES 

Use the RENAME statement to rename the equivalent numeric variables to the original character variable names: 

rename x1=earnq3; 
rename x2=earnq4; 

Note that a variable’s type, character or numeric, cannot be changed once the variable is created. A new SAS data 
set needs to be created to store the equivalent numeric variables. Steps 2-4 can be consolidated in one DATA step: 

data earnings2; 
set earnings; 
x1=input(earnq3,best12.); 
x2=input(earnq4,best12.); 
drop earnq3 earnq4;  
rename x1=earnq3; 
rename x2=earnq4; 
run; 

Also note that the equivalent numeric variables are added after the original variables and do not have labels:  

CONTENTS of data set EARNINGS2 
 
#    Variable    Type    Len    Label 
 
1    id          Char      2    Unique identifier     
2    earnq1      Num       8    Earnings in Quarter 1 
3    earnq2      Num       8    Earnings in Quarter 2 
4    employed    Char      3    Ever employed         
5    earnq3      Num       8                          
6    earnq4      Num       8   

 

STEP 5: REORDERING VARIABLES TO THE ORIGINAL SEQUENCE 

Use the RETAIN statement to reorder variables to the original sequence: 

data earnings3; 
retain id earnq1-earnq4 employed; 
set earnings2; 
run; 

Note that SAS adds variables to the data set being created in the order in which it encounters them. Placing the 
RETAIN statement after the DATA statement, but before the SET statement, effectively adds the variables named 
on the RETAIN statement to the data set being created in the specified order. Numbered range lists of variables 
(EARNQ1-EARNQ4) can be used. 

SAS picks up the variable attributes (type and length, plus format and/or label, if assigned) when it executes the 
SET statement and reads the descriptor portion of data set EARNINGS2. 
 

STEP 6: LABELING THE EQUIVALENT NUMERIC VARIABLES 

Use the LABEL statement to label the equivalent numeric variables with the original character variable labels: 

label earnq3='Earnings in Quarter 3'; 
label earnq4='Earnings in Quarter 4'; 

Note that a new SAS data set needs to be created to reorder variables. Variables can be labeled with the LABEL 
statement at the same time. Steps 5-6 can be consolidated in one DATA step: 

data earnings3; 
retain id earnq1-earnq4 employed; 
set earnings2; 
label earnq3='Earnings in Quarter 3'; 
label earnq4='Earnings in Quarter 4'; 
run; 

Character variables that stored numbers have now been converted to numeric variables with the same names, 
labels, and order: 
 

CONTENTS of data set EARNINGS3 
 
#    Variable    Type    Len    Label 
 
1    id          Char      2    Unique identifier     
2    earnq1      Num       8    Earnings in Quarter 1 
3    earnq2      Num       8    Earnings in Quarter 2 
4    earnq3      Num       8    Earnings in Quarter 3 
5    earnq4      Num       8    Earnings in Quarter 4 
6    employed    Char      3    Ever employed 

PRINT of data set EARNINGS3 
 
Obs    id    earnq1    earnq2    earnq3    earnq4    employed 
 
 1     01     1000      1300      1400      1250       1     
 2     02        0      2000         .      2300       1     
 3     03        0         0         0         0       UNK     
 4     04     1400      1400      1500       900       1     
 5     05        0         .         0         0       UNK 

 

The above six steps produce the desired result but require coding that would be prohibitive with a larger number 
of variables. The next section describes how to generate the same code using SAS. 

CONVERTING VARIABLES AUTOMATICALLY 
 

STEP 1: IDENTIFYING CHARACTER VARIABLES TO CONVERT 

First, identify character variables in SAS data set EARNINGS. Use PROC SQL to retrieve their names from a 
DICTIONARY table. 

DICTIONARY.COLUMNS stores the names and types of all variables in all data sets with one variable per row. 
Do not retrieve ID; although it stores only numeric values, they contain leading zeroes that should be preserved: 

%let excludevars=id; 
 

proc sql noprint; 
select name into :charvars separated by ' ' 
from dictionary.columns 
where libname="WORK" and memname="EARNINGS" and type="char" 
      and not indexw(upcase("&excludevars"),upcase(name)); 
quit; 

The %LET statement assigns the value ID to a macro variable named EXCLUDEVARS. 

The PROC SQL statement starts the procedure. The NOPRINT option suppresses printed output. 

The SELECT clause picks up values of NAME (variable name). The INTO clause stores values in a macro variable 
named CHARVARS, separated by blanks. 

The FROM clause retrieves rows from DICTIONARY.COLUMNS. 

The WHERE clause subsets rows where four conditions are met: 

1. the value of LIBNAME equals “WORK” 
2. the value of MEMNAME equals “EARNINGS” 
3. the value of TYPE equals “char” 
4. the value of NAME is not in the list of variables to be excluded 

(Note: Values of LIBNAME and MEMNAME are stored in uppercase and values of TYPE are stored in lowercase.) 

In other words, the WHERE clause subsets rows for character variables in SAS data set WORK.EARNINGS that 
are not specifically excluded.  

The INDEXW function returns the starting position of the second argument if it is found in the first argument or 0 
if it is not found. Variable names may be in uppercase, lowercase, or mixed case; the UPCASE function is used on 
both arguments to compare values in the same case. When the expression is NOT true (i.e., when the result is 0, 
meaning the variable name was not found in the list of variables to exclude), the row is selected. 

Second, determine the number of character variables: 

%let ncharvars=%sysfunc(countw(&charvars)); 

The %LET statement uses the COUNTW function to count the number of words (i.e., variable names) stored in macro 
variable CHARVARS. COUNTW is a DATA step function; it must be called by %SYSFUNC in a %LET statement. 

The values of macro variables CHARVARS and NCHARVARS can be inspected with %PUT:  

%put charvars=&charvars; 
%put ncharvars=&ncharvars; 

SAS writes the following to the log: 
 

charvars=earnq3 earnq4 employed 
ncharvars=3 

 

Third, use the DATA step to check variables for non-numeric values. This includes three techniques: 

1. using the INPUT function to convert character variables to numeric variables 
2. keeping running totals of non-numeric values with accumulator variables 
3. assigning the names of variables to a macro variable with CALL SYMPUTX 
 
The result is a macro variable named VARLIST that stores the names of variables to be converted.  

data _null_; 
set earnings end=lastobs; 
array charvars{*} &charvars; 
array charvals{&ncharvars}; 
do i=1 to &ncharvars; 
  if input(charvars{i},?? best12.)=. and charvars{i} ne ' ' then charvals{i}+1; 
end; 
if lastobs then do; 
  length varlist $ 32767; 
  do j=1 to &ncharvars; 
    if charvals{j}=. then varlist=catx(' ',varlist,vname(charvars{j})); 
  end; 
  call symputx('varlist',varlist); 
end; 
run; 

The DATA _NULL_ statement allows DATA step processing but does not create a SAS data set. 

The SET statement reads observations from SAS data set EARNINGS. The END= option creates a temporary 
variable named LASTOBS that is initialized to 0. LASTOBS equals 1 when SET reads the last observation in 
SAS data set EARNINGS. 

The first ARRAY statement creates an array named CHARVARS for EARNQ3, EARNQ4, and EMPLOYED (i.e.,  
the value of macro variable CHARVARS). 

The second ARRAY statement creates an array named CHARVALS with 3 (i.e., the value of macro variable 
NCHARVARS) variables. No variables are specified, so SAS creates CHARVALS1 through CHARVALS3. 
No $ is specified, so SAS creates numeric variables. The variables are initialized to missing values. 

The DO loop initializes index variable I, which iterates from 1 to 3 (i.e., the value of macro variable NCHARVARS). 
EARNQ3 is processed first, followed by EARNQ4 and EMPLOYED: 

First, it checks if converting the value of EARNQ3 to a number produces a missing value. It uses the INPUT 
function to read the value of character variable EARNQ3 with the numeric informat BEST12. The result is a 
numeric value or, if the value of EARNQ3 is non-numeric, a missing value. (The ?? after the comma prevents 
printing an error message in the log and setting the automatic variable _ERROR_ to 1 if the value of EARNQ3 
is non-numeric.) 

Second, it checks if the value of EARNQ3 does not equal missing. If EARNQ3 is missing, the INPUT function 
will, appropriately, convert it to a missing numeric value. It should not be counted as a non-numeric value. 

If the expression is true (i.e., converting EARNQ3 to a numeric value produces a missing value and EARNQ3 
itself is not missing), it adds 1 to the value of the ith variable in array CHARVALS. CHARVALS{I}+1 is a SUM 
statement. It creates accumulator variables whose values are automatically set to 0 before SAS reads the first 
observation and then retained across observations (i.e., creates running totals). 

When the SET statement reads the last observation in SAS data set EARNINGS, the value of temporary variable 
LASTOBS equals 1. (IF LASTOBS is equivalent to IF LASTOBS=1.) SAS runs the statements in the DO loop. 

The LENGTH statement creates a character variable named VARLIST that is 32767 bytes long (i.e., the maximum 
length of a character variable). 

The DO loop initializes index variable J, which iterates from 1 to 3 (i.e., the value of macro variable NCHARVARS). 

The IF statement checks if the jth variable in array CHARVALS is missing. Each variable in array CHARVALS stores 
the number of values of the relative variable in array CHARVARS that could not be converted to numeric values. If 
the jth variable in array CHARVALS is missing, there was no value that could not be converted to a numeric value 
(i.e., the jth variable in array CHARVARS can be converted to a numeric variable). 

The CATX function has three arguments. The first argument is a blank specified in quotes. It will delimit the values 
of subsequent arguments. The second argument is VARLIST, which was initialized to missing. The third argument 
uses the VNAME function to return the name of the respective variable in array CHARVARS that can be converted 
to a numeric variable. As the DO loop iterates, the CATX function continuously adds to the list of variables. 

Finally, CALL SYMPUTX is used to assign the value of the variable VARLIST to a macro variable named VARLIST. 
This macro variable can be used after this DATA step. 

The value of macro variable VARLIST can be inspected with %PUT:  

%put varlist=&varlist; 

SAS writes the following to the log: 
 

varlist=earnq3 earnq4 
 

STEP 2: CREATING EQUIVALENT NUMERIC VARIABLES 
Use the INPUT function in an assignment statement to convert a character variable to a numeric variable. Define 
separate arrays to store the character variables and the numeric variables. Use a DO loop to process them: 

%let nvars=%sysfunc(countw(&varlist)); 
 
data earnings2; 
set earnings; 
array charx{&nvars} &varlist; 
array     x{&nvars}         ; 
do i=1 to &nvars; 
  x{i}=input(charx{i},best12.); 
end; 
run; 

The %LET statement uses the COUNTW function to count the number of words (i.e., variable names) stored in 
macro variable VARLIST. 

The DATA statement creates SAS data set EARNINGS2. 

The SET statement reads observations from SAS data set EARNINGS. 

The first ARRAY statement creates an array named CHARX with 2 (i.e., the value of macro variable NVARS) 
variables, EARNQ3 and EARNQ4 (i.e., the value of macro variable VARLIST). 

The second ARRAY statement creates an array named X with 2 variables. No variables are specified, so SAS 
creates X1 and X2. No $ is specified, so SAS creates numeric variables. 

The DO loop initializes index variable I, which iterates from 1 to 2 (i.e., the value of macro variable NVARS). 

The assignment statement reads the ith character variable in array CHARX with the numeric informat BEST12., 
converts it to a numeric value, and stores the result in the ith numeric variable in array X. 

The END; statement ends the DO loop. 

STEP 3: DROPPING THE ORIGINAL CHARACTER VARIABLES 

Use the DROP statement to drop the original character variables that stored numbers. Reference the macro 
variable VARLIST which stores the names of the original character variables. Also drop I, the index variable: 

drop &varlist i; 
 

STEP 4: RENAMING THE EQUIVALENT NUMERIC VARIABLES 

Use the RENAME statement to rename the equivalent numeric variables to the original character variable names. 
Note that the names of variables to be converted are stored in macro variable VARLIST. The equivalent numeric 
variable names are in the sequence X1-Xn. The number of variables to be converted is stored in macro variable 
NVARS. This is enough information to automate renaming with a macro. 

Use a macro %DO loop to write RENAME statements: 

%macro renamer; 
  %do i=1 %to &nvars; 
    rename x&i = %scan(&varlist,&i) ; 
  %end; 
%mend renamer; 

The %MACRO statement defines a macro named RENAMER. 

The %DO loop initializes index variable I, which iterates from 1 to 2 (i.e., the value of macro variable NVARS). 

The next line is passed as text. The RENAME statement renames Xn (where n is the current value of index 
variable I) to the ith word in the list of variables to be converted. X1 is renamed to EARNQ3, X2 to EARNQ4.  

The %END; statement ends the %DO loop. 

The %MEND statement ends the macro definition. 

Step 2 created equivalent numeric variables. Step 3 dropped the original character variables. Step 4 renamed 
the equivalent numeric variables. All of these take place in the DATA step. Steps 2-4 can be consolidated: 

%let nvars=%sysfunc(countw(&varlist)); 
 
%macro renamer; 
  %do i=1 %to &nvars; 
    rename x&i = %scan(&varlist,&i) ; 
  %end; 
%mend renamer; 
 
data earnings2; 
set earnings; 
array charx{&nvars} &varlist; 
array     x{&nvars}         ; 
do i=1 to &nvars; 
  x{i}=input(charx{i},best12.); 
end; 
drop &varlist i; 
%renamer 
run; 

Note that the macro definition for RENAMER precedes the DATA step and is called at the end of the DATA step. 
Variables are renamed as the final action in the DATA step (i.e., after the original character variables of the same 
name are dropped). 

Use OPTIONS MPRINT; to display the code generated by executing the macro (optional): 
 

MPRINT(RENAMER):   rename x1 = earnq3 ; 
MPRINT(RENAMER):   rename x2 = earnq4 ; 

Note that the equivalent numeric variables are added after the original variables and do not have labels: 
 

CONTENTS of data set EARNINGS2 
 
#    Variable    Type    Len    Label 
 
1    id          Char      2    Unique identifier     
2    earnq1      Num       8    Earnings in Quarter 1 
3    earnq2      Num       8    Earnings in Quarter 2 
4    employed    Char      3    Ever employed         
5    earnq3      Num       8                          
6    earnq4      Num       8   

 

STEP 5: REORDERING VARIABLES TO THE ORIGINAL SEQUENCE 

Use the RETAIN statement to reorder variables to the original sequence. Retrieve the names of variables from 
DICTIONARY.COLUMNS: 

proc sql noprint; 
select name into :orderlist separated by ' ' 
from dictionary.columns 
where libname="WORK" and memname="EARNINGS" 
order by varnum; 
quit; 

The PROC SQL statement starts the procedure. The NOPRINT option suppresses printed output. 

The SELECT clause picks up values of NAME (variable name). The INTO clause stores values in a macro variable 
named ORDERLIST, separated by blanks. 

The FROM clause reads rows from DICTIONARY.COLUMNS. 

The WHERE clause subsets rows where the value of LIBNAME equals “WORK”  and the value of MEMNAME 
equals “EARNINGS” (i.e., rows for variables in SAS data set WORK.EARNINGS). 

The ORDER BY clause orders rows by VARNUM (variable creation order). 

The value of macro variable ORDERLIST can be inspected with %PUT:  

%put orderlist=&orderlist; 

SAS writes the following to the log: 
 

orderlist=id earnq1 earnq2 earnq3 earnq4 employed 
 

Use macro variable ORDERLIST on a RETAIN statement to reorder variables: 

data earnings3; 
retain &orderlist; 
set earnings2; 
run; 

Placing the RETAIN statement after the DATA statement, but before the SET statement, adds the variables on 
the RETAIN statement (i.e., the names stored in macro variable ORDERLIST) to SAS data set EARNINGS3 in 
the specified order. 
 

STEP 6: LABELING THE EQUIVALENT NUMERIC VARIABLES 

Use the LABEL statement to label the equivalent numeric variables with the original character variable labels. 
Information on variable labels is also stored in DICTIONARY.COLUMNS. Use PROC SQL to store LABEL 
statements in a macro variable: 

proc sql noprint; 
select catx(' ','label',name,'=',quote(trim(label)),';') 
       into :labels separated by ' ' 
from dictionary.columns 
where libname="WORK" and memname="EARNINGS" and 
      indexw(upcase("&varlist"),upcase(name)); 
quit; 

The PROC SQL statement starts the procedure. The NOPRINT option suppresses printed output. 

The SELECT clause builds a string with the CATX function. The INTO clause stores values (one LABEL statement 
for each variable) in a macro variable named LABELS, separated by blanks. 

The CATX function has six arguments. The first argument is a blank specified in quotes. It will delimit the values 
of subsequent arguments. The remaining arguments are concatenated into one string, each a respective LABEL 
statement. Each string starts with the word ‘label’ followed by the variable name, an equals sign, and the variable 
label. The TRIM function removes any trailing blanks from the value of LABEL. The QUOTE function encloses the 
(trimmed) value of LABEL in double quotes. Each string ends with a semicolon. 

The FROM clause reads rows from DICTIONARY.COLUMNS. 

The WHERE clause subsets rows where the value of LIBNAME equals “WORK” and the value of MEMNAME equals 
“EARNINGS” (i.e., rows for variables in SAS data set WORK.EARNINGS) and the value of NAME (variable name) is 
in the list of variables to be converted. 

The value of macro variable LABELS can be inspected with %PUT:  

%put labels=%quote(&labels); 

SAS writes the following to the log: 
 

labels=label earnq3 = "Earnings in Quarter 3" ; label earnq4 = "Earnings in 
Quarter 4" ; 

 

(The value of macro variable LABELS contains a semicolon at the end of each LABEL statement. Use the macro 
quoting function %QUOTE to treat the value as text.) 

Macro variable LABELS can be used to label the equivalent numeric variables. 

Step 5 reordered variables to the original sequence. Step 6 created a macro variable that stores labels for the 
equivalent numeric variables. Both steps use PROC SQL to create macro variables that can be referenced in 
a DATA step. Steps 5-6 can be consolidated: 

proc sql noprint; 

select name into :orderlist separated by ' ' 
from dictionary.columns 
where libname="WORK" and memname="EARNINGS" 
order by varnum; 
 
select catx(' ','label',name,'=',quote(trim(label)),';') 
       into :labels separated by ' ' 
from dictionary.columns 
where libname="WORK" and memname="EARNINGS" and 
      indexw(upcase("&varlist"),upcase(name)); 
 
quit; 

data earnings3; 
retain &orderlist; 
set earnings2; 
&labels 
run; 

Character variables that stored numbers have now been converted to numeric variables with the same names, 
labels, and order: 
 

CONTENTS of data set EARNINGS3 
 
#    Variable    Type    Len    Label 
 
1    id          Char      2    Unique identifier     
2    earnq1      Num       8    Earnings in Quarter 1 
3    earnq2      Num       8    Earnings in Quarter 2 
4    earnq3      Num       8    Earnings in Quarter 3 
5    earnq4      Num       8    Earnings in Quarter 4 
6    employed    Char      3    Ever employed 

 

PRINT of data set EARNINGS3 
 
Obs    id    earnq1    earnq2    earnq3    earnq4    employed 
 
 1     01     1000      1300      1400      1250       1     
 2     02        0      2000         .      2300       1     
 3     03        0         0         0         0       UNK     
 4     04     1400      1400      1500       900       1     
 5     05        0         .         0         0       UNK 

 

The above still used some hard-coding (e.g., data set names). A macro to fully automate this process is included 
in the Appendix. 

CONCLUSION 
SAS can be used to identify character variables that store numbers and to create a data set with equivalent 
numeric variables with the same names, labels, and order. 

APPENDIX 
The following macro converts character variables that store numbers to numeric variables. Results are saved in 
the specified output data set. Converted variables have the same names, labels, and order: 
 

%macro char2num(inputlib=work,     /* libref for input data set  */ 
                inputdsn=,         /* name of input data set     */ 
                outputlib=work,    /* libref for output data set */ 
                outputdsn=,        /* name of output data set    */ 
                excludevars=);     /* variables to exclude       */ 
 
proc sql noprint; 
select name into :charvars separated by ' ' 
from dictionary.columns 
where libname=upcase("&inputlib") and memname=upcase("&inputdsn") and type="char" 
      and not indexw(upcase("&excludevars"),upcase(name)); 
quit; 
 
%let ncharvars=%sysfunc(countw(&charvars)); 
 
data _null_; 
set &inputlib..&inputdsn end=lastobs; 
array charvars{*} &charvars; 
array charvals{&ncharvars}; 
do i=1 to &ncharvars; 
  if input(charvars{i},?? best12.)=. and charvars{i}  ne ' ' then charvals{i}+1; 
end; 
if lastobs then do; 
  length varlist $ 32767; 
  do j=1 to &ncharvars; 
    if charvals{j}=. then varlist=catx(' ',varlist,vname(charvars{j})); 
  end; 
  call symputx('varlist',varlist); 
end; 
run; 
 
%let nvars=%sysfunc(countw(&varlist)); 
 
data temp; 
set &inputlib..&inputdsn; 
array charx{&nvars} &varlist; 
array     x{&nvars}         ; 
do i=1 to &nvars; 
  x{i}=input(charx{i},best12.); 
end; 
drop &varlist i; 
 
%do i=1 %to &nvars; 
  rename x&i = %scan(&varlist,&i) ; 
%end; 
 
run; 

proc sql noprint; 
 
select name into :orderlist separated by ' ' 
from dictionary.columns 
where libname=upcase("&inputlib") and memname=upcase("&inputdsn") 
order by varnum; 
 
select catx(' ','label',name,'=',quote(trim(label)),';')  
       into :labels separated by ' ' 
from dictionary.columns 
where libname=upcase("&inputlib") and memname=upcase("&inputdsn") and  
      indexw(upcase("&varlist"),upcase(name)); 
 
quit; 
 
data &outputlib..&outputdsn; 
retain &orderlist; 
set temp; 
&labels 
run; 
 
%mend char2num; 

 
 
Sample macro call: 

 
%char2num(inputdsn=earnings,      /* read WORK.EARNINGS   */ 
          outputdsn=earningsC,    /* write WORK.EARNINGSC */ 
          excludevars=id)         /* do not convert ID    */ 

 
 
INPUTLIB and OUTPUTLIB default to WORK (i.e., temporary SAS data sets). 
 
INPUTDSN and OUTPUTDSN are required parameters. 
 
EXCLUDEVARS is optional. Specify variables that should not be converted separated by at least one space. 
Variable names may be in uppercase, lowercase, or mixed case. 
 
Note: This macro is not guaranteed to work in every situation. 

